Algorithms for MWA Graduation Formulas

Eric S. Seah*

Department of Actuarial and Management Sciences

University of Manitoba

Abstract

Shiu recently used matrix theory to derive some formulas for calculating the coefficients of minimum-R_Z moving-weighted-average formulas. In this note, we give an *APL* implementation for one of Shiu's formulas.

1. Introduction

For many years, actuaries have studied and used the method of graduation by moving weighted averages, which is included in the syllabus of course 165 of the Society of Actuaries examinations (see [6], Chapter 3). Matrix derivation of the coefficients of the symmetric and asymmetric minimum-R_z moving-weighted-average (MWA) formulas is, however, a fairly recent development. In 1947 and 1948, Dr. T.N.E. Greville [1, 2] determined the coefficients of some MWA formulas and presented these coefficients in matrix form. Later on, he used the matrix approach to solve for a specific case [3]. Recently, Dr. E.S.W. Shiu [7, 8] generalized this approach and derived some elegant formulas. In this note, we present *APL* programs to carry out the computation for one of the formulas derived by Shiu.

* Support from the University Research Committee of Senate, University of Manitoba, is gratefully acknowledged.

2. Notation

Consider the N-term MWA formula

$$v_x = \sum_{i=-k}^{m} c_i . u_{x+i} \tag{2.1}$$

where k and m are non-negative integers and $k + 1 + m = N$. Note that when $k = m$, we have the symmetric formulas. Let $\mathbf{c} = (c_{-k}, ..., c_0, ..., c_m)^T$, and $\mathbf{j} = (0, ..., 0, 1, 0, ..., 0)^T$, where there are k 0's above and m 0's below the 1 in $\mathbf{j}$.

Let $\mathbf{K}_{z,n}$ denote the $(n-z)$-row by n-column differencing matrix, the ij-th entry of which is given by

$$(-1)^{z+i-j} \binom{z}{j-i}, \text{ where } \binom{z}{k} = 0 \text{ for } k < 0 \text{ and } k > z.$$

Let $\{ p_0, p_1, ..., p_r \}$ be a basis for the linear space of all polynomials of degree r or less. For example, let $p_i(x) = x^i$. Define $\mathbf{P}$ to be the N-row by $(r+1)$-column matrix such that the $(j+1)$ th column of $\mathbf{P}$ equals to $(p_j(-k), ..., p_j(0), ..., p_j(m))^T$.

3. Matrix derivation of the coefficients of MWA formulas

The two criteria used to determine the coefficients $\{c_i\}$ in (2.1) are fitness and smoothness.

The graduated data $\{v_i\}$ is said to have a reasonable fit to the ungraduated data $\{u_i\}$, if (2.1) is exact for polynomials up to a certain degree, say r. That is, the coefficients $\{c_i\}$ are such that, for each polynomial p of degree r or less,

$$p(y) = \sum_{i=-k}^{m} c_i \cdot p(y+i) \qquad (3.1)$$

for all y, but (3.1) is not true in general for polynomials of degree greater than r.

For smoothness, we require that, for a given positive integer z, the z th differences of the sequence of graduated values $\{ v_i \}$ are numerically small.

The MWA formula derived with these criteria is the one that is classically referred to as minimum-R_z.

Using these two criteria and matrix theory, Shiu ([7], pp. 489-491) showed that the coefficients of the minimum-R_z MWA formula are given by the entries in the column vector $\mathbf{c} = \mathbf{A}\mathbf{j}$. Here, $\mathbf{A}$ is a N-row by N-column matrix and is given by

$$\mathbf{A} = (\mathbf{K}\mathbf{K}^T)^{-1} \mathbf{P} (\mathbf{P}^T (\mathbf{K}\mathbf{K}^T)^{-1} \mathbf{P})^{-1} \mathbf{P}^T, \qquad (3.2)$$

where $\mathbf{K} = \mathbf{K}_{z,\, N+z}$. Note that (3.2) requires two matrix inversions. In [7], Shiu gave an *APL* program for (3.2).

Shiu [8] derived another formula for the matrix $\mathbf{A}$:

$$\mathbf{A} = \mathbf{I} - \mathbf{K}^T_{r+1,\, N} \, (\mathbf{K}_{t,\, N+z} \mathbf{K}^T_{t,\, N+z})^{-1} \mathbf{K}_{t,\, N+z} \mathbf{K}^T_{z,\, N+z} \qquad (3.3)$$

where $t = r+1+z$. With this formula, only one matrix inversion is needed.

4. An algorithm for calculating the coefficients of MWA formulas

Let $\mathbf{G} = \mathbf{K}_{t,N+z} \mathbf{K}^T_{t,N+z}$, the matrix to be inverted in (3.3). We remark that $\mathbf{G}$ is a band-diagonal matrix of bandwidth $2t + 1$, with the binomial coefficients in the expansion of $(-1)^t(x-1)^{2t}$ displayed symmetrically about the main diagonal in each row and each column. Note that G is a square matrix of dimension $(N-r-1)$. Shiu [8] gave a construction of a non-singular upper triangular matrix U, where GU is lower triangular. Consequently, $G^{-1} = U(U^TGU)^{-1}U^T$, where U^TGU is diagonal and its inverse can be easily calculated. As pointed out by Shiu, the inverse of G has been derived by Hoskins and Ponzo [5].

In fact, Hoskins and Ponzo proved that the ij-th entry of the inverse of $(-1)^tG$ is given by:

$$(-1)^t \binom{i+t-1}{t}\binom{j+t-1}{t}\sum_{l=i}^{q}\frac{\binom{l+t-i-1}{t-1}\binom{l+t-j-1}{t-1}}{\binom{l+t-1}{t}\binom{l+2t-1}{t}}, \quad (3.4)$$

where $q = N-r-1$.

We implemented (3.3) in *APL*, using (3.4) with the omission of $(-1)^t$ to compute the inverse of G. This approach has the advantage that we can skip the intermediate steps of computing U, U^TGU and $(U^TGU)^{-1}$. The *APL* programs are listed in the appendix. As an example, to obtain the coefficients of all five-term minimum-R_3 exact-for-cubics MWA formula, we enter

MWA 3 3 5,

where the first 3 refers to the minimum-R_3, the second 3 is associated with the third degree of exactness, and 5 is the number of terms in the MWA formula. The computer returns the matrix A (rounded to 5 decimal places):

0.96503	0.06119	−0.07343	0.06119	−0.03497
0.13986	0.75524	0.29371	−0.24476	0.13986
−0.20979	0.36713	0.55944	0.36713	−0.20979
0.13986	−0.24476	0.29371	0.75524	0.13986
−0.03497	0.06119	−0.07343	0.06119	0.96503

For example, the coefficients of the asymmetric five-term minimum-R_3 exact-for-cubics

MWA formula with $k = 1$ and $m = 3$ are given by the second column of $\boldsymbol{A}$ (cf. [7], p. 491). This matrix is also found in Greville ([2], p. 13).

References

1. Greville, T.N.E., "Actuarial Note: Adjusted Average Graduation Formulas of Maximum Smoothness," *The Record of the American Institute of Actuaries* 36 (1947), 249-264.

2. Greville, T.N.E., "Actuarial Note: Tables of Coefficients in Adjusted Average Graduation Formulas of Maximum Smoothness," *The Record of the American Institute of Actuaries* 37 (1948), 11-30.

3. Greville, T.N.E., "On Smoothing a Finite Table: A Matrix Approach," *Journal of the Society for Industrial and Applied Mathematics* 5 (1957), 137-154. Reprinted in [4].

4.. Greville, T.N.E., *Selected Papers of T.N.E. Greville.* Charles Babbage Research Centre, Manitoba, Canada (1984).

5. Hoskins, W.D. and P.J. Ponzo, "Some Properties of a Class of Band Matrices," *Mathematics of Computation* 26 (1972), 393-400.

6. London, D., *The Revison of Estimates.* Actex Publications, Winstead, CT (1985).

7. Shiu, E.S.W., "Minimum-R_Z Moving-Weighted-Average Formulas," *TSA* 36 (1984), 489-500.

8. Shiu, E.S.W., "Matrix Derivation of Moving-Weighted-Average Graduation Formulas," *Insurance: Mathematics and Economics* 6 (1987), 1-6.

Appendix

```
    ∇ C←MWA RZN;R;Z;N;I;A;K1;K2;K3;K4
      R←RZN[1]
      Z←RZN[2]
      N←RZN[3]
      I←(⍳N)∘.=⍳N
      K1←⍉(R+1) DIF N
      K2←(R+Z+1) INV(N-R)-1
      K3←(R+Z+1) DIF N+Z
      K4←⍉Z DIF N+Z
      C←I-K1+.×K2+.×K3+.×K4
    ∇
```

```
    ∇ K←Z DIF N;V
      V←(1+N)↑((0,⍳Z)!Z)×⌽(Z+1)⍴ 1 ¯1
      K←((N-Z),N)⍴V
    ∇
```

```
    ∇ K←T INV N;F1;F2;F3;F4;I;J
      F1←T!(T-1)+⍳N
      F2←((⍳N)∘.≤⍳N)×(T-1)!((N,N)⍴⍳N)+(T-1)-⍉(N,N)⍴⍳N
      F3←T!(⍳N)+T-1
      F4←T!(⍳N)+T+T-1
      K←(N,N)⍴0
      I←1
     LP1:J←I
     LP2:K[J;I]←K[I;J]←F1[I]×F1[J]×+/F2[I;]×F2[J;]÷F3×F4
      →(N≥J←J+1)/LP2
      →(N≥I←I+1)/LP1
    ∇
```